2006 Data Users Conference – NAMCS/NHAMCS Hands On Exercises

SAS Exercises:

```
*********************************************************************;
/*Exercise 1 - Reading 2004 NAMCS public use data*/
/*Read NAMCS data & create a SAS dataset called namcsduc */

filename namcs04 'c:\namcs\duc\namcs04'; /*data file*/

filename nam04inp 'c:\namcs\duc\nam04inp.txt'; /*input*/
filename nam04for 'c:\namcs\duc\nam04for.txt'; /*formats*/
filename nam04lab 'c:\namcs\duc\nam04lab.txt'; /*lables*/

%inc nam04for;

data namcsduc;
infile namcs04 missover lrecl=999;
%inc nam04inp;
%inc nam04lab;

patwt1k=patwt/1000;
keep phycode patcode ager sex diag13d patwt patwt1k cpsum cstratm gen1-
gen8
specr timemd;
run;

proc print data=namcsduc (obs=5);
title '2004 NAMCS Visits-Selected Variables';
run;

*********************************************************************;
/* Exercise 2-Simple Frequencies, unweighted & weighted */
/* Estimate the number of office visits by sex & age */

proc freq data=namcsduc;
tables sex*ager;
title 'Exercise 2a-Unweighted 2004 NAMCS Visits: Sex by Age';
run;

proc freq data=namcsduc;
tables sex*ager /list;
weight patwt1k;
title 'Exercise 2b-Weighted 2004 NAMCS Visits: Sex by Age';
run;
```

```
*************************************************************************;
/*Exercise 3- Simple Frequencies with standard errors, SURVEYFREQ */
/*Estimate the number of office visits by sex & age with SEs */
proc surveyfreq data=namcsduc;
tables sex*ager /clwt cl;
cluster cpsum;
strata cstratm;
weight patwt1k;
TITLE 'Exercise 3-2004 NAMCS Visits-Sex by Age ';
run;

*************************************************************************;
/*Exercise 4 - Transforming data, SURVEYFREQ */
/*Breakdown of primary dx-listed asthma visits by sex*/
data namex4;
set namcsduc;
if DIAG13D='493' then asthma=1;
else asthma=2;
proc surveyfreq data=namex4;
tables sex*asthma /row;
cluster cpsum;
strata cstratm;
weight patwt1k;
TITLE 'Exercise 4 - 2004 NAMCS Asthma Visits by Sex';
run;
*************************************************************************;
/*Exercise 5 - Creating asthma visit rates, SURVEYFREQ*/
data namex5;
set namex4;
totrt=(patwt/288375857)*100;
if sex=1 then sexwt=(patwt/147412657)*100;
if sex=2 then sexwt=(patwt/140963200)*100;

/* 5a -  Total asthma visits per 100 persons...note - this will give
you the correct visit rate per 100 persons, but incorrect rate per 100
male/female */
proc surveyfreq data=namex5;
tables asthma*sex /clwt;
cluster cpsum;
strata cstratm;
weight totrt;
Title 'Exercise 5a - 2004 NAMCS Asthma Visits per 100 persons';
run;

/* 5b - Total asthma visits per 100 male/female...note - this will give
you the correct visit rate per 100 male/female, but the incorrect visit
rate per 100 persons. */
proc surveyfreq data=namex5;
tables asthma*sex /clwt;
cluster cpsum;
strata cstratm;
weight sexwt;
Title 'Exercise 5b - 2004 NAMCS Asthma Visits per 100 male/female';
run;
```

```
*************************************************************************;
/*Exercise 6-Estimate total number of Ibuprofen (gen=52730) mentions*/
data namex6;
set namcsduc;
************************;
total=0;
************************;
array meds(8) gen1 gen2 gen3 gen4 gen5 gen6 gen7 gen8;
do j=1 to 8;
if meds(j)= '52730' then total=total+1;
end;
proc surveymeans data=namex6 nobs sum mean;
cluster cpsum;
strata cstratm;
var total;
weight patwt1k;
Title "Exercise 6 - 2004 NAMCS Ibuprofen Mentions";
run;

*************************************************************************;
/*Exercise 7-Mean number of minutes spent with physician*/
data namex7;
set namcsduc;
if timemd=0 then timemd=.; /*used to delete timemd=0 visits*/
proc surveymeans data=namex7 nobs sum stderr mean;
domain specr;
class specr;
cluster cpsum;
strata cstratm;
var timemd;
weight patwt1k;
Title "Exercise 7 - 2004 NAMCS Time Spent with Physician by Specialty";
run;
*************************************************************************;
```

STATA Exercises:

```
/* make the weights more manageable for viewing */
gen patwt1k=patwt/1000

/* change delimiter for the long line of code to follow*/
# delimit;

/* keeping the variables of interest for this exercise*/
keep phycode patcode ager sex diag13d patwt patwt1k cpsum cstratm
gen1-gen8 specr timemd;
# delimit cr

/*********************************************************************/
/* Exercise 2 & 3 - Simple frequencies, unweighted and weighted */

tab sex ager

svyset cpsum [pweight=patwt1k], strata(cstratm)

svy: tab sex ager, count format(%10.0f) cellwidth(12) se

/*********************************************************************/
/* Exercise 4 - Transforming data */

gen asthma=0
replace asthma=1 if diag13d=="493"
svyset cpsum [pweight=patwt1k], strata(cstratm)
svy: tab sex asthma, count format(%10.0f) se

/* logistic regression */
svy: logistic asthma sex ager

/* show OR for each age group */
xi:svy: logistic asthma sex i.ager

/* change the reference group */
char ager[omit] 6
xi:svy: logistic asthma sex i.ager


/*********************************************************************/
/* Exercise 5 - Creating asthma visit rates */

gen totrt=(patwt/288375857)*100
gen sexwt=(patwt/147412657)*100 if sex==1
replace sexwt=(patwt/140963200)*100 if sex==2


/* 5a -  Total asthma visits per 100 persons...note - this will give
you                */
/*       the correct visit rate per 100 persons, but incorrect rate per
100 male/female */

svyset cpsum [pweight=totrt], strata(cstratm)
svy: tab asthma sex, count se ci
```

```
/* 5b - Total asthma visits per 100 male/female...note - this will give
you the correct visit rate per 100 male/female, but the incorrect visit
rate per 100 persons. */

svyset cpsum [pweight=sexwt], strata(cstratm)
svy: tab asthma sex, count se ci


/*****************************************************************/
/*Exercise 6-Estimate total number of Ibuprofen (gen=52730) mentions*/

gen totmed=0
foreach var of varlist gen1 gen2 gen3 gen4 gen5 gen6 gen7 gen8 {
      replace totmed=totmed+1 if `var'==52730
}
svyset cpsum [pweight=patwt1k], strata(cstratm)
svy: mean totmed
svy: total totmed


/*****************************************************************/
/* Exercise 7 - Mean number of minutes spent with physician */

svyset cpsum [pweight=patwt1k], strata(cstratm)
replace timemd=. if timemd==0
svy:mean timemd, over(specr)
/*****************************************************************/
```

SUDAAN Exercises:

```
*********************************************************************;
/*Exercise 1 - Reading 2004 NAMCS public use data*/
/*Read NAMCS data & create a SAS dataset called namcsduc */

filename namcs04 'c:\namcs\duc\namcs04'; /*data file*/

filename nam04inp 'c:\namcs\duc\nam04inp.txt'; /*input*/
filename nam04for 'c:\namcs\duc\nam04for.txt'; /*formats*/
filename nam04lab 'c:\namcs\duc\nam04lab.txt'; /*lables*/

%inc nam04for;

data namcsduc;
infile namcs04 missover lrecl=999;
%inc nam04inp;
%inc nam04lab;

patwt1k=patwt/1000;
keep phycode patcode ager sex diag13d patwt patwt1k cpsum cstratm gen1-
gen8
specr timemd;
run;
*********************************************************************
/*Exercise 3-Simple Frequencies with standard errors-SUDAAN */
/*Estimate the number of office visits by sex & age with SEs*/;
proc sort data=namcsduc;
by cstratm cpsum;
proc crosstab data=namcsduc filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt1k;
class sex ager;
tables sex*ager;
print nsum wsum sewgt totper setot/ nsumfmt=f8 wsumfmt=f8 sewgtfmt=f8.1
colperfmt=f8.1 secolfmt=f8.1
style=nchs;
run;
*********************************************************************;
/*Exercise 4 - Transforming data, SURVEYFREQ */
/*Breakdown of primary dx-listed asthma visits by sex*/
data namex4;
set namcsduc;
if DIAG13D='493' then asthma=1;
else asthma=2;
proc sort data=namex4;
by cstratm cpsum;
proc crosstab data=namex4 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt1k;
subpopn asthma=1;
class sex ager;
tables sex*ager;
print nsum wsum sewgt totper setot rowper serow /
nsumfmt=f8. wsumfmt=f10 sewgtfmt=f8 totperfmt=f8.1
rowperfmt=f8.1 serowfmt=f8.1  style=nchs;
run;
```

```
*************************************************************************;
/*Exercise 5 - Creating asthma visit rates, SURVEYFREQ*/
/*Estimate number of visits for asthma per 100 persons*/
data namex5;
set namex4;
totrt=(patwt/288375857)*100;
if sex=1 then sexwt=(patwt/147412657)*100;
if sex=2 then sexwt=(patwt/140963200)*100;
proc sort data=namex5;
by cstratm cpsum;
/* 5a -  Total asthma visits per 100 persons...note - this will give
you the correct visit rate per 100 persons, but incorrect rate per 100
male/female */
proc crosstab data=namex5 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight totrt;
class asthma sex;
tables asthma*sex;
print nsum wsum sewgt  /
wsumfmt=f10.1 sewgtfmt=f8.1 nsumfmt=f8.
style=nchs;
run;
/* 5b - Total asthma visits per 100 male/female...note - this will give
you the correct visit rate per 100 male/female, but the incorrect visit
rate per 100 persons. */
proc crosstab data=namex5 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight sexwt;
class asthma sex;
tables asthma*sex;
print nsum wsum sewgt  /
wsumfmt=f10.1 sewgtfmt=f8.1 nsumfmt=f8.
style=nchs;
run;
*************************************************************************;
/*Exercise 6-Estimate total number of Ibuprofen (gen=52730)
mentions*/data namex6;
data namex6;
set namcsduc;
************************;
total=0;
************************;
array meds(8) gen1 gen2 gen3 gen4 gen5 gen6 gen7 gen8;
do j=1 to 8;
if meds(j)= '52730' then total=total+1;
end;
proc sort data=namex6;
by cstratm cpsum;
proc descript data=namex6 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt1k;
var total;
print mean semean total setotal nsum /
totalfmt=f13.3 setotalfmt=f10.3 nsumfmt=f8.
style=nchs;
run;
```

```
*************************************************************************;
/*Exercise 7-Mean number of minutes spent with physician*/
data namex7;
set namcsduc;
proc sort data=namex7;
by cstratm cpsum;
proc descript data=namex7 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt1k;
subpopn timemd>0;
class specr;
var timemd;
print mean semean nsum /
totalfmt=f10.3 setotalfmt=f8.3 nsumfmt=f8.
style=nchs;
run;
*************************************************************************;
```